

Formalising the Local Compactness of the Adele Ring

Salvatore Mercuri   

The University of Edinburgh, UK

Abstract

The adèle ring of a number field is a central object in modern number theory. Its status as a locally compact topological ring is one of the key reasons why, leading to its widespread use within the Langlands Program. We describe a formal proof that the adèle ring of a number field is locally compact in the Lean 4 theorem prover. Our work includes the formalisations of new types, including the completion of a number field at an infinite place and the finite S -adèle ring, as well as formal proofs that completions of a number field are locally compact and their rings of integers at finite places are compact.

2012 ACM Subject Classification Theory of computation → Logic and verification; Theory of computation → Type theory

Keywords and phrases formal mathematics, algebraic number theory, Lean, mathlib

Supplementary Material

Source code: https://github.com/smercuri/adele-ring_locally-compact/tree/journal

Documentation: https://smercuri.github.io/adele-ring_locally-compact/

Acknowledgements I would like to thank María Inés de Frutos-Fernández for their advice, in particular for highlighting that the infinite adèle ring of a number field could now be formalised as the product of Archimedean completions.

1 Introduction

In number theory, looking locally can give a lot of information about the global picture with which we are primarily concerned. For example, we might be interested in finding integer or rational solutions to the equation $x^2 + 3y^2 = 2024$. It turns out there are none, because any rational solution would lead to a local solution modulo 3 of the equation $x^2 \equiv 2 \pmod{3}$, yet 2 is not a square modulo 3. Without this local information, this fact is a lot more difficult to prove. This is a powerful tool that can work both ways; for example, under certain conditions Hasse’s *local-global principle* provides a global rational solution of an equation if there is a real solution and local solutions for every prime p .

The local study of the field \mathbb{Q} of rational numbers at a prime p is systematised in the field \mathbb{Q}_p of p -adic numbers. Just as the real numbers \mathbb{R} is the completion of \mathbb{Q} with respect to the usual absolute value, the field \mathbb{Q}_p is the completion of \mathbb{Q} with respect to the p -adic absolute value. In line with the local-global principle, intuitively we might then unify all real and local information into the product $\mathbb{R} \times \prod_p \mathbb{Q}_p$, where p ranges over all primes. The adèle ring $\mathbb{A}_{\mathbb{Q}}$ of the rationals is a subring of this product defined by a restriction on the infinite product over p which we detail later. This particular restriction makes the adèle ring a *locally compact* topological ring.

While the adèle ring is useful as a tool for encoding local information, its status as a locally compact topological ring has led to its widespread use in modern number theory. One particular example is the ability to do Fourier analysis on the adèle ring, which Tate used in their thesis [10] to reformulate Dirichlet L -functions and their functional equations on adèles. The normalisation factors that were previously inserted somewhat artificially into the functional equations to make them work now arise naturally from the definition of the

adelic Dirichlet L -function at the real component of the rational adèle ring. Of course, Tate's thesis is really the theory of automorphic forms on $GL(1)$ and a precursor to the Langlands program where the adèle ring analogously plays a foundational role.

Adele rings of general number fields can be defined similarly to the rational adèle ring. In this paper, we formalise the proof that the adèle ring of a number field is a locally compact topological ring in the Lean theorem prover, v4.7.0, with the use of Lean's mathematics library `mathlib` [6]. This is a milestone along the long-term path of formalising Tate's thesis and the Langlands program within `mathlib`. Prior work [4] formalised the adèle ring of a global field in Lean; additions made to `mathlib` since then now enable us to formalise of the completion of a number field at its infinite places, and we use this to provide an alternative formalisation of the infinite adèle ring of a number field. With the help of recent work formalising local fields by [5], we also formalise important and necessary topological results, such as the proof that the local ring of integers is compact.

We provide some mathematical and brief `mathlib` preliminaries in Section 2, involving the definitions of absolute values, discrete valuations, number fields, and locally compact topological spaces, as well as some theory of the uniform space completion operation. Section 3 concerns the formalisation of the adèle ring of a number field; in particular, the new formalisations of the completion of a number field at an infinite place and the infinite adèle ring is detailed in Section 3.1. We then describe the informal and formal proof that the adèle ring is locally compact, along with other topological results, in Section 4. Section 5 concludes with a discussion around implementation details.

2 Preliminaries

2.1 Absolute values and discrete valuations

While absolute values and discrete valuations are often unified into a single treatment, we separate them here due to their distinction within `mathlib`. See [3, Chapter 2] and [9, Chapter 1] for details on the theory of absolute values and discrete valuations respectively, and [5, Section 2] for details of integer-valued multiplicative valuations in `mathlib`.

► **Definition 2.1** (Absolute value). *Let K be a field. We say that the function $|\cdot| : K \rightarrow \mathbb{R}_{\geq 0}$ is an absolute value on K if, for all $x, y \in K$, we have:*

1. $|x| = 0$ if and only if $x = 0$;
2. $|xy| = |x||y|$;
3. $|x + y| \leq |x| + |y|$ (the triangle inequality).

If, for all $x, y \in K$, the ultrametric inequality,

4. $|x + y| \leq \max(x, y)$,

holds over the triangle inequality, the absolute value is non-Archimedean; else, it is Archimedean.

Discrete *additive* valuations on K are group homomorphisms $K^\times \rightarrow (\mathbb{Z}, +)$, satisfying certain properties and extended to all of K as follows.

► **Definition 2.2** (Discrete additive valuation). *The function $a : K \rightarrow \mathbb{Z} \cup \{\infty\}$ is a discrete additive valuation on K if, for all $x, y \in K$, we have:*

1. $a(x) = \infty$ if and only if $x = 0$;
2. $a(xy) = a(x) + a(y)$;
3. $a(x + y) \geq \min(a(x), a(y))$.

► **Example 2.3.** If $K = \mathbb{Q}$, then the additive p -adic valuation a_p (defined on an integer as the largest power of p dividing that integer; extended to \mathbb{Q} by $a_p(a/b) := a_p(a) - a_p(b)$) is an example of a discrete additive valuation.

We can construct a non-Archimedean absolute value from a discrete additive valuation a by defining $|x|_a = c^{-a(x)}$ if $x \neq 0$ and $|0|_a = 0$, for some chosen real number $c > 1$. The p -adic absolute value $|x|_p = p^{-a_p(x)}$ is an example of this construction when $K = \mathbb{Q}$.

Discrete valuations in the literature are typically defined as additive valuations. In `mathlib`, discrete *multiplicative* valuations have a more developed API than their additive counterparts. These are abstractions of the absolute values arising from discrete additive valuations that do not require a choice of the base c . In `mathlib`, any abstract additive structure can be turned into a corresponding abstract multiplicative structure, which is in bijection with the original additive type and preserves preorders. For example, the additive group of integers $(\mathbb{Z}, +)$ bijects with the type \mathbb{Z}_m , which can be thought of as the set \mathbb{Z} equipped with multiplication, via the following inverse morphisms:

$$\begin{aligned} \text{ofAdd} &: (\mathbb{Z}, +) \rightarrow \mathbb{Z}_m; \\ \text{toAdd} &: \mathbb{Z}_m \rightarrow (\mathbb{Z}, +). \end{aligned}$$

In particular, these maps preserve respective units, generators and preorders, so $\text{ofAdd}(0) = 1$, $\text{ofAdd}(1)$ generates \mathbb{Z}_m , and $\text{ofAdd}(x) \leq \text{ofAdd}(y)$ if and only if $x \leq y$. This is analogous to the way in which we turn the additive structure of \mathbb{Z} to a multiplicative one via exponentiation. To transfer infinite additive values to the multiplicative setting we add a zero term to obtain the type \mathbb{Z}_{m0} , which extends the multiplicative structure and the preorder of \mathbb{Z}_m .

► **Definition 2.4** (Discrete multiplicative valuation). *The function $v : K \rightarrow \mathbb{Z}_{m0}$ is a discrete multiplicative valuation on K if, for all $x, y \in K$, we have:*

1. $v(0) = 0$;
2. $v(1) = 1$;
3. $v(xy) = v(x)v(y)$;
4. $v(x + y) \leq \max\{v(x), v(y)\}$.

We emphasise that a discrete multiplicative valuation is not the same as the absolute value $|\cdot|_a$ constructed from a discrete additive valuation a . The former only takes values in \mathbb{Z} , while the latter does not; the latter depends on a choice of base c , while the former does not. However, each additive valuation a also gives rise to a multiplicative valuation v by defining $v(x) = \text{ofAdd}(-a(x))$ for $x \neq 0$ and $v(0) = 0$.

► **Example 2.5.** If $K = \mathbb{Q}$, then the p -adic valuation v_p is defined by $v_p(x) = \text{ofAdd}(-a_p(x))$ if $x \neq 0$. We see that x is a p -adic unit if and only if $v_p(x) = \text{ofAdd}(0) = 1$, and $v_p(x) \leq 1$ if and only if $a_p(x) \geq 1$.

► **Definition 2.6.** *Two absolute values (resp. discrete multiplicative valuations) $|\cdot|_1$ and $|\cdot|_2$ (resp. v_1 and v_2) on K are equivalent if $|\cdot|_1 = |\cdot|_2^\alpha$ (resp. $v_1 = v_2^\alpha$) for some $\alpha \in \mathbb{R}_{>0}$.*

► **Definition 2.7.** *An infinite place of a field K is an equivalence class of Archimedean absolute values on K . A finite place of K is an equivalence class of discrete multiplicative valuations on K .*

We typically care more about *places* of a field K over individual valuations, because equivalent valuations define the same topology on K and lead to the same completion of the field K .

2.2 Number fields

A number field K is a finite-degree field extension of \mathbb{Q} , and its ring of integers \mathcal{O}_K is the integral closure of \mathbb{Z} inside \mathbb{Q} . If K is not a unique factorisation domain, then the fundamental theorem of arithmetic does not hold at the level of elements, however it does at the level of ideals. That is, any non-zero ideal of \mathcal{O}_K can be written as a finite product of non-zero prime ideals, unique up to reordering of the prime ideals.

► **Definition 2.8** (\mathfrak{p} -adic valuation). *Given a non-zero prime ideal \mathfrak{p} of \mathcal{O}_K , the additive \mathfrak{p} -adic valuation $a_{\mathfrak{p}}$ on $x \in \mathcal{O}_K$ is the largest power of \mathfrak{p} appearing in the factorisation of $x\mathcal{O}_K$ into prime ideals. This extends to a discrete additive valuation on K by defining $a_{\mathfrak{p}}(a/b) = a_{\mathfrak{p}}(a) - a_{\mathfrak{p}}(b)$ and $a_{\mathfrak{p}}(0) = \infty$. The \mathfrak{p} -adic valuation $v_{\mathfrak{p}}$ is the multiplicative valuation defined as $v_{\mathfrak{p}}(x) = \text{ofAdd}(-a_{\mathfrak{p}}(x))$ if $x \neq 0$ and $v_{\mathfrak{p}}(0) = 0$.*

2.3 Uniform spaces and their completions

While topological spaces abstract the notion of continuity, *uniform spaces*, first introduced by [12], provide the appropriate abstraction of *uniform* continuity. We describe some brief relevant theory here; for more information, we refer the reader to [2, Sections 2, 5]. A uniform space (X, \mathcal{U}) consists of a set X and a filter \mathcal{U} on $X \times X$ satisfying certain properties. Uniform continuity of a map $f : (X, \mathcal{U}_X) \rightarrow (Y, \mathcal{U}_Y)$ between uniform spaces is understood via the condition that $\mathcal{U}_X \leq (f \times f)^* \mathcal{U}_Y$, where $(f \times f)^* \mathcal{U}_Y$ is the *pullback* filter generated by preimages of elements in \mathcal{U}_Y under f .

Uniform spaces come with a powerful completion operation, which makes use of the unified notion of limits that filters represent. Much like metric spaces can be completed by adding limits of Cauchy sequences, there is an operation sending a uniform space (X, \mathcal{U}) to a complete separated uniform space $(\widehat{X}, \widehat{\mathcal{U}})$ by adding limits of Cauchy *filters*. This completion operator satisfies the universal property of [1, Theorem II.3.7.3, Definition II.3.7.4] (see also [2, Theorem 5.1]), hence it defines a functor from the category of all uniform spaces to the category of complete separated uniform spaces. The universal property of the uniform completion functor means that it represents an equivalence class of *abstract completion operators* as follows.

► **Definition 2.9** ([2]). *We say that $(Y, \iota : X \rightarrow Y)$ is an abstract completion of the uniform space X if Y is a complete separated uniform space, ι has dense image and ι is uniform continuous.*

► **Theorem 2.10** ([2]). *Any two abstract completions $(Y_i, \iota_i : X \rightarrow Y_i)$, for $i = 1, 2$, of a uniform space X have the property that Y_1 and Y_2 are isomorphic as uniform spaces.*

These properties make the uniform completion operation ideal for formalising the completions of a range of objects, including the number fields that we consider within this paper. This is implemented within `mathlib` alongside a robust API as `UniformSpace.Completion`, which is an instance of the abstract completion class given by `AbstractCompletion`.

A field K can have various sources of uniform structures. An absolute value $|\cdot|$ on K gives rise to a metric space via the distance $d(x, y) := |x - y|$ and therefore a uniform structure through the filter generated by the sets $\{(x, y) \mid d(x, y) < \varepsilon\}$, where $\varepsilon > 0$. Discrete multiplicative valuations v similarly define uniform spaces (K, \mathcal{U}_v) . In this way we can obtain the completion K_v of a field at each place v via the uniform space completion functor.

► **Example 2.11.** 1. The uniform structure $(\mathbb{Q}, \mathcal{U}_{\infty})$ arising from the usual absolute value $|\cdot|_{\infty}$ on \mathbb{Q} defines a uniform space, whose completion is the field \mathbb{R} of real numbers.

2. The uniform structure $(\mathbb{Q}, \mathcal{U}_{v_p})$ arising from the multiplicative p -adic valuation defines a uniform space, whose completion is the field \mathbb{Q}_p of p -adic numbers. The subring \mathbb{Z} of \mathbb{Q} completes to the p -adic ring of integers $\mathbb{Z}_p := \{x \in \mathbb{Q}_p \mid v_p(x) \leq 1\}$.

If (Y, \mathcal{U}_Y) is a uniform space and $f : X \rightarrow Y$ is a function on sets, then the pullback filter $(f \times f)^* \mathcal{U}_Y$ defines the *pullback uniform structure* on X . In `mathlib`, this pullback structure is implemented as `UniformSpace.comap`. This operation can provide other sources of uniform structures on a field K should it embed into another uniform field L , for example.

In general, the completion \widehat{K} of a separated uniform space (K, \mathcal{U}) may not be a field, even if K is a field. The necessary and sufficient condition required to guarantee this is that (K, \mathcal{U}) is a *completable topological field*, given in [1, Theorem III.6.8.7] and also [2, Theorem 5.4], which requires the inverse map $x \mapsto x^{-1}$ on K to preserve the Cauchy property of filters which do not have a cluster point at zero. The type class `CompletableTopField` encodes this property within `mathlib`.

The uniform space (K, \mathcal{U}_v) arising from a discrete multiplicative valuation is a completable topological field, which is currently implemented in `mathlib` as `Valued.completableTopField`. The pullback $(\sigma \times \sigma)^* \mathcal{U}_L$ uniform structure on K under a field embedding $\sigma : K \hookrightarrow L$ also defines a completable topological field, provided that (L, \mathcal{U}_K) is a completable topological field. This is because σ is uniform continuous and the inversion map commutes with σ . We implement this as follows.

```
theorem UniformSpace.comap_completableTopField {α : Type u} {β : Type v}
  [Field β] [Field α] {f : α →+* β} [b : UniformSpace β]
  [@TOSpace α (UniformSpace.comap f b).toTopologicalSpace]
  [CompletableTopField β] :
  @CompletableTopField _ _ (UniformSpace.comap f b) := by ...
```

2.4 Locally compact space

There are many characterisations of a topological space being locally compact. We shall use the following notion, which is implemented in `mathlib` through the type class `LocallyCompactSpace`.

► **Definition 2.12** (Locally Compact). *A topological space X is locally compact if, for each $x \in X$ and each neighbourhood N of x , there exists a compact neighbourhood $S \subseteq N$ of x .*

It is well known that closed subspaces of locally compact spaces, finite products of locally compact spaces, and infinite products of compact spaces are all locally compact.

3 The adèle ring of a number field

Ostrowski's theorem [7] states that the only infinite place of \mathbb{Q} is represented by the usual absolute value $|\cdot|_\infty$ and all the finite places are represented by the p -adic valuations v_p . The *infinite adèle ring* $\mathbb{A}_{\mathbb{Q}, \infty}$ of \mathbb{Q} is the completion \mathbb{R} of \mathbb{Q} at the single infinite place, and the *finite adèle ring* $\mathbb{A}_{\mathbb{Q}, f}$ is the restricted product of the completions of \mathbb{Q} over primes p ,

$$\mathbb{A}_{\mathbb{Q}, f} := \left\{ (x_p)_p \in \prod_p \mathbb{Q}_p \mid x_p \in \mathbb{Z}_p \text{ for all but finitely many } p \right\}. \quad (1)$$

The adèle ring of \mathbb{Q} is then the product of the infinite and finite adèle rings, $\mathbb{A}_{\mathbb{Q}} = \mathbb{A}_{\mathbb{Q}, \infty} \times \mathbb{A}_{\mathbb{Q}, f}$. The story is similar for general number fields, with the infinite places coming from the complex absolute value $|\cdot|_{\mathbb{C}}$ and the finite places being represented by \mathfrak{p} -adic valuations of non-zero prime ideals \mathfrak{p} of \mathcal{O}_K .

3.1 The infinite adèle ring of a number field

```
namespace NumberField.InfinitePlace
```

Throughout Section 3.1 the following variables are in scope.

```
variable (K : Type*) [Field K] [NumberField K] (v : NumberField.InfinitePlace K)
```

3.1.1 Infinite places of a number field

There is only a single infinite place of \mathbb{Q} because there is only a single field embedding of \mathbb{Q} into \mathbb{C} . For a general number field, we get a finite number of additional embeddings coming from the Galois action permuting roots of minimal polynomials over \mathbb{Q} . Distinct embeddings $\sigma : K \hookrightarrow \mathbb{C}$ give non-equivalent Archimedean absolute values on K via $|x|_\sigma := |\sigma(x)|_{\mathbb{C}}$. These define all infinite places of a number field, the set of which is denoted by $\Sigma_{K,\infty}$.

Infinite places of a number field K were first added to `mathlib` by [8] as the subtype, `NumberField.InfinitePlace K`, of absolute values on K coming from the composition of a complex embedding of K with the canonical norm on \mathbb{C} . Associated to any infinite place v is an absolute value `v.val : AbsoluteValue K ℝ` and an embedding `v.embedding : K →+* ℂ` such that `(norm : ℂ → ℝ).comp v.embedding = v.val`.

3.1.2 Completing a number field at an infinite place

Each infinite place $v \in \Sigma_{K,\infty}$ defines a uniform structure \mathcal{U}_v on K through its associated absolute value and we can therefore complete K at v to obtain K_v , as described in Section 2.3. There are many ways to formalise the uniform structure \mathcal{U}_v . For example, we could use the uniform space instance `AbsoluteValue.uniformSpace` of the absolute value `v.val` associated to the infinite place. Alternatively we can leverage the fact that this absolute value factors through an embedding into \mathbb{C} , and use this to pullback the uniform structure of \mathbb{C} . This approach has the advantage of allowing us to transfer useful instances from \mathbb{C} . For example, (K, \mathcal{U}_v) is separated and a completable topological field because \mathbb{C} is.

```
instance uniformSpace : UniformSpace K :=
  UniformSpace.comap v.embedding inferInstance

instance uniformAddGroup : @UniformAddGroup K v.uniformSpace _ :=
  UniformAddGroup.comap v.embedding

instance t0Space : @T0Space K v.topologicalSpace :=
  @t0Space_of_injective_of_continuous _ _ v.topologicalSpace _ _
  v.embedding.injective v.embedding_continuous _

instance completableTopField :
  @CompletableTopField K _ v.uniformSpace := UniformSpace.comap_completableTopField
```

With these instances, we can formalise the field K_v with a topological ring structure.

```
def completion := @UniformSpace.Completion K v.uniformSpace

namespace Completion

instance : Field (v.completion K) := @UniformSpace.Completion.instField _ _
  v.uniformSpace v.topologicalDivisionRing _ _
```

```
instance : TopologicalRing (v.completion K) :=
  @UniformSpace.Completion.topologicalRing K _ v.uniformSpace _ v.uniformAddGroup
end InfinitePlace.Completion
```

3.1.3 Formalising the infinite adele ring

► **Definition 3.1** (Infinite adele ring). *The infinite adele ring $\mathbb{A}_{K,\infty}$ of a number field K is the product topological ring over all infinite place completions of K :*

$$\mathbb{A}_{K,\infty} := \prod_{v \in \Sigma_{K,\infty}} K_v. \quad (2)$$

```
def infiniteAdeleRing := (v : NumberField.InfinitePlace K) → v.completion K

namespace InfiniteAdeleRing

instance : TopologicalRing (infiniteAdeleRing K) := Pi.instTopologicalRing

end NumberField.InfiniteAdeleRing
```

3.2 The finite adele ring of a number field

The finite adele ring was formalised in `mathlib` by [4] for any Dedekind domain R of Krull dimension one and its field of fractions K , by taking the restricted product over the non-zero prime spectrum $\text{Spec}(R)$ of R . Formally, we work in the same generality. However, for simplicity our informal discussion restricts to $R = \mathcal{O}_K$, as required by our main results on number fields. Throughout Section 3.3 the following variables are in scope.

```
variable (R : Type*) [CommRing R] [IsDomain R] [IsDedekindDomain R]
         (K : Type*) [Field K] [Algebra R K] [IsFractionRing R K]
```

3.2.1 Finite places of a number field

Prior work of [4, 5] formalised discrete valuations, finite places of global fields and completions thereof. The set of all finite places of a number field K , denoted by $\Sigma_{K,f}$, is indexed by the non-zero prime ideals $\mathfrak{p} \in \text{Spec}(\mathcal{O}_K)$ through their \mathfrak{p} -adic valuations. The type `WithZero (Multiplicative ℤ)` represents the multiplicative integers \mathbb{Z}_{m0} , and the structure `Valuation K (WithZero (Multiplicative ℤ))` represents discrete multiplicative valuations of K . The typing `v : IsDedekindDomain.HeightOneSpectrum (ringOfIntegers K)` represents the relation $\mathfrak{p} \in \text{Spec}(\mathcal{O}_K)$ and we can access the associated multiplicative valuation of any such prime ideal via `v.valuation : Valuation K (WithZero (Multiplicative ℤ))`. Moreover, each prime ideal term `v` defines a uniform structure on K , via `v.adicValued.toUniformSpace`, which is then used to formalise the v -adic completion K_v of K as `v.adicCompletion K`. The v -adic ring of integers, $\mathcal{O}_v := \{x \in K_v \mid v(x) \leq 1\}$, is the type `v.adicCompletionIntegers K`.

3.2.2 Formalising the finite adele ring

```
namespace DedekindDomain.FiniteAdeleRing
```

► **Definition 3.2** (Finite adèle ring). *The finite adèle ring $\mathbb{A}_{K,f}$ of a number field K ,*

$$\mathbb{A}_{K,f} := \left\{ (x_v)_v \in \prod_{v \in \Sigma_{K,f}} K_v \mid x_v \in \mathcal{O}_v \text{ for all but finitely many } v \right\}, \quad (3)$$

is the restricted product topological ring whose topology is generated by the following basis:

$$\left\{ \prod_{v \in \Sigma_{K,f}} V_v \mid V_v \subseteq K_v \text{ open and } V_v = \mathcal{O}_v \text{ for all but finitely many } v \right\}.$$

While the type, `DedekindDomain.finiteAdeleRing R K`, encoding the finite adèle ring is now available within `mathlib`, the topology of [4] is not, so we adapt this as follows.

```
def generatingSet : Set (Set (finiteAdeleRing R K)) :=
  Set.preimage (Subtype.val) '' (Set.pi Set.univ '' (
    {V | (∀ v, IsOpen (V v)) ∧
      (∀f v in Filter.cofinite, V v = v.adicCompletionIntegers K)}))

instance topologicalSpace : TopologicalSpace (finiteAdeleRing R K) :=
  TopologicalSpace.generateFrom (generatingSet R K)
```

The finite adèle ring defines a topological ring if addition and multiplication are continuous. The proof of this was formalised in [4], which we abstract to the result that any function on $\mathbb{A}_{K,f} \times \mathbb{A}_{K,f}$ is continuous if it factors through certain continuous functions at each place. This establishes the topological ring structure of the finite adèle ring.

► **Theorem 3.3.** *Let $f : \mathbb{A}_{K,f} \times \mathbb{A}_{K,f} \rightarrow \mathbb{A}_{K,f}$ and $g_v : K_v \times K_v \rightarrow K_v$ be functions such that $f(x, y)_v = g_v(x_v, y_v)$ for all $v \in \Sigma_{K,f}$. If all the g_v are continuous, and $g_v(\mathcal{O}_v \times \mathcal{O}_v) \subseteq \mathcal{O}_v$, then f is continuous.*

```
theorem continuous_if_factors2
  (f : finiteAdeleRing R K × finiteAdeleRing R K → finiteAdeleRing R K)
  (hf : ∃ g : (v : HeightOneSpectrum R) →
    v.adicCompletion K × v.adicCompletion K → v.adicCompletion K,
    ∀ v, (g v ∘ (RingHom.prodMap (π v) (π v))) = (π v) ∘ f ∧
    Continuous (g v) ∧ ∀ (x y : v.adicCompletion K),
      x ∈ v.adicCompletionIntegers K → y ∈ v.adicCompletionIntegers K
      → g v (x, y) ∈ v.adicCompletionIntegers K) :
    Continuous f := ...

end DedekindDomain.FiniteAdeleRing
```

3.3 The adèle ring

► **Definition 3.4** (Adèle ring). *The adèle ring \mathbb{A}_K of a number field is the product topological ring of the infinite and the finite adèle rings:*

$$\mathbb{A}_K := \mathbb{A}_{K,\infty} \times \mathbb{A}_{K,f}. \quad (4)$$

```
def NumberField.adèleRing (K : Type*) [Field K] [NumberField K] :=
  infiniteAdeleRing K × finiteAdeleRing (ringOfIntegers K) K
```


4 Local compactness of the adèle ring of a number field

This section contains an informal and formal proof that the adèle ring of a number field is locally compact, using [3, Chapter 2] as source. Full details of the formal proof can be found in the source code. The informal proofs are written so as to align with the formal proofs.

4.1 Local compactness of the infinite adèle ring

```
namespace NumberField.InfinitePlace
```

Throughout Section 4.1 the following variables are in scope.

```
variable (K : Type*) [Field K] [NumberField K] (v : NumberField.InfinitePlace K)
```

► **Theorem 4.1.** *The completion K_v of a number field K at $v \in \Sigma_{K,\infty}$ is locally compact.*

Proof. The uniform embedding $\sigma : K \hookrightarrow \mathbb{C}$ extends to an embedding $\sigma_v : K_v \hookrightarrow \mathbb{C}$ by the universal property of uniform completions. This embedding has a closed image inside a locally compact space, so K_v is locally compact. ◀

That the embedding $\sigma : K \hookrightarrow \mathbb{C}$ is a uniform embedding follows immediately from the facts that σ is injective and that the uniform space on K is the pullback along σ . Further, σ becomes an isometry if we assign K a compatible pseudometric space instance.

```
theorem uniformEmbedding : @UniformEmbedding _ _ v.uniformSpace _ v.embedding := by
  rw [@uniformEmbedding_iff, @uniformInducing_iff_uniformSpace]
  exact ⟨rfl, v.embedding.injective⟩

theorem uniformInducing : @UniformInducing _ _ v.uniformSpace _ v.embedding :=
  @UniformEmbedding.toUniformInducing _ _ v.uniformSpace _ _ v.uniformEmbedding

instance pseudoMetricSpace : PseudoMetricSpace K :=
  @UniformInducing.comapPseudoMetricSpace _ _ v.uniformSpace _ _ v.uniformInducing

theorem isometry :
  @Isometry _ _ v.pseudoMetricSpace.toPseudoEMetricSpace _ (v.embedding) :=
  @UniformEmbedding.to_isometry _ _ v.uniformSpace _ _ v.uniformEmbedding
```

Next, we prove that the isometric property of σ transfers through to its extension σ_v , which guarantees that σ_v has a closed image. We use the fact that σ is an isometry and that K has dense image in K_v to show that σ_v preserves distances.

```
namespace Completion

def extensionEmbedding := @UniformSpace.Completion.extensionHom K _ v.uniformSpace
  v.topologicalRing v.uniformAddGroup _ _ _ _ v.embedding v.continuous _ _

theorem extensionEmbedding_dist_eq (x y : v.completion K) :
  dist (extensionEmbedding K v x) (extensionEmbedding K v y) = dist x y := ...

theorem extensionEmbedding_isometry : @Isometry _ _
  (metricSpace K v).toPseudoEMetricSpace _ (extensionEmbedding K v) :=
  Isometry.of_dist_eq extensionEmbedding_dist_eq
```

```

theorem closedEmbedding : ClosedEmbedding (extensionEmbedding K v) :=
  (extensionEmbedding_isometry K v).closedEmbedding

theorem locallyCompactSpace : LocallyCompactSpace (v.completion K) :=
  (closedEmbedding K v).locallyCompactSpace

end InfinitePlace.Completion

```

► **Theorem 4.2.** *The infinite adèle ring $\mathbb{A}_{K,\infty}$ of a number field K is locally compact.*

Proof. Since each completion K_v is locally compact by Theorem 4.1 then, as a finite product of locally compact spaces, so is $\mathbb{A}_{K,\infty}$. ◀

```

namespace InfiniteAdeleRing

theorem locallyCompactSpace : LocallyCompactSpace (infiniteAdeleRing K) :=
  Pi.locallyCompactSpace_of_finite

end NumberField.InfiniteAdeleRing

```

4.2 Local compactness of the finite adèle ring

Showing the local compactness of the finite adèle ring is difficult due to it being an infinite product. The first step is to understand the *compactness* of each v -adic ring of integers \mathcal{O}_v and the *local compactness* of each v -adic completion K_v . Next, we construct an open cover of the finite adèle ring consisting of the finite S -adèle rings, which are much easier to understand as locally compact, and from this the local compactness of the finite adèle ring can be deduced. Throughout Section 4.2 the following variables are in scope.

```

variable (R : Type*) [CommRing R] [IsDedekindDomain R]
  (K : Type*) [Field K] [Algebra R K] [IsFractionRing R K]
  (v : HeightOneSpectrum R) (S : Finset (HeightOneSpectrum R))

```

4.2.1 Compactness of the v -adic ring of integers

```

namespace IsDedekindDomain.HeightOneSpectrum.AdicCompletionIntegers

```

The v -adic ring of integers is a local ring whose unique maximal ideal \mathfrak{m}_v we alias as follows.

```

def maximalIdeal : Ideal (v.adicCompletionIntegers K) :=
  LocalRing.maximalIdeal (v.adicCompletionIntegers K)

```

► **Theorem 4.3.** *The ring of integers \mathcal{O}_v of the v -adic completion K_v of a number field K at a finite place v is compact.*

Proof. We show that \mathcal{O}_v is complete and totally bounded, which ensures that it is compact. \mathcal{O}_v is a closed set of the complete space K_v , hence complete. It is totally bounded if, for any $\gamma \in \mathbb{Z}_m$, there exists a finite cover of \mathcal{O}_v of open balls $B_\gamma(t) := \{x \mid v(x - t) < \gamma\}$ of radius γ . If $\gamma > 1$ then the single open ball $B_\gamma(0)$ covers \mathcal{O}_v . Else, if $\gamma = 1$, then take the finitely-many representatives t_i of the residue field $\mathcal{O}_v/\mathfrak{m}_v$; else, if $\gamma < 0$, then take the finitely-many representatives t_i of $\mathcal{O}_v/\mathfrak{m}_v^{-\text{toAdd}(\gamma)+1}$. The balls $B_\gamma(t_i)$ cover \mathcal{O}_v . ◀

```

theorem isTotallyBounded :
  TotallyBounded (v.adicCompletionIntegers K : Set (v.adicCompletion K)) := ...

instance : CompleteSpace (v.adicCompletionIntegers K) :=
  IsClosed.completeSpace_coe (isClosed K v)

theorem isCompact :
  IsCompact (v.adicCompletionIntegers K : Set (v.adicCompletion K)) :=
  isCompact_iff_totallyBounded_isComplete.2
  ⟨isTotallyBounded K v, IsClosed.isComplete (isClosed K v)⟩

```

The proof of the above requires that the residue field $\mathcal{O}_v/\mathfrak{m}_v$ is *finite*; this is not currently in `mathlib`, but it is available in the work of [5] and hence we leave it as an open `sorry` in our work. We extend this to show that $\mathcal{O}_v/\mathfrak{m}_v^n$ is finite, for any integer $n \geq 0$, by defining the finite-coefficient map `toFiniteCoeffs n hπ` sending $x \in \mathcal{O}_v/\mathfrak{m}_v^n$ to an n -tuple (x_1, \dots, x_n) of v -adic digits in its π -adic expansion, where π is a uniformizer, and showing that this defines an injective function $\mathcal{O}_v/\mathfrak{m}_v^n \rightarrow (\mathcal{O}_v/\mathfrak{m}_v)^n$.

```

theorem toFiniteCoeffs_injective {π : v.adicCompletionIntegers K} (n : ℕ)
  (hπ : IsUniformizer π.val) : (toFiniteCoeffs n hπ).Injective := ...

theorem quotient_maximalIdeal_pow_finite {π : v.adicCompletionIntegers K} (n : ℕ)
  (hπ : IsUniformizer π.val) :
  Fintype (v.adicCompletionIntegers K / (maximalIdeal K v)^n) :=
  Fintype.ofInjective _ (toFiniteCoeffs_injective n hπ)

```

► **Theorem 4.4.** *The v -adic completion K_v of a number field at $v \in \Sigma_{K,f}$ is locally compact.*

Proof. Let $x \in K_v$. Each neighbourhood of x contains an open ball $B_\gamma(x)$ centred at x for some $\gamma > 0$, and each such open ball contains the continuous image of \mathfrak{m}_v via translation and dilation. This image is compact, since \mathfrak{m}_v is compact, and it is a neighbourhood of x . ◀

```

instance locallyCompactSpace : LocallyCompactSpace (v.adicCompletion K) := ...

end IsDedekindDomain.HeightOneSpectrum.AdicCompletionIntegers

```

4.2.2 The finite S -adele ring

```

namespace DedekindDomain

```

Associated to any element x of the finite adele ring is a finite set of places $S_x \subseteq \Sigma_{K,f}$ for which $x_v \in \mathcal{O}_v$ if and only if $v \notin S_x$. It is the dependence of the set S_x on x that makes the local compactness of the finite adele ring difficult to view. On the other hand, if we fix a finite set $S \subseteq \Sigma_{K,f}$, and consider only the finite adeles x for which $S_x \subseteq S$, then this is an easier space to understand as locally compact. These are the finite S -adeles.

► **Definition 4.5** (Finite S -adele ring). *Let $S \subseteq \Sigma_{K,f}$ be a finite set of places of a number field K . The finite S -adele ring $\mathbb{A}_{K,S,f}$ is the topological ring, defined as a subring of $\prod_v K_v$ as*

$$\mathbb{A}_{K,S,f} := \left\{ (x_v)_v \in \prod_{v \in \Sigma_{K,f}} K_v \mid x_v \in \mathcal{O}_v \text{ for all } v \notin S \right\}.$$

While $\mathbb{A}_{K,S,f}$ is given the subspace topology of $\prod_v K_v$, it can also be viewed as a subspace of $\mathbb{A}_{K,f}$, so it has two immediate sources of topologies. These define the *same* topology on $\mathbb{A}_{K,S,f}$, which is a crucial step in the overall proof that $\mathbb{A}_{K,f}$ is locally compact.

For any Dedekind domain R of Krull dimension one and its field of fractions K , we formalise the finite S -adele ring as a term of type `Subring (ProdAdicCompletions R K)`, as opposed to `Subring (finiteAdeleRing R K)`. This turns out to be the right hierarchy in which to place $\mathbb{A}_{K,S,f}$, as it streamlines the formal proof that $\mathbb{A}_{K,S,f}$ is locally compact significantly.

```
def IsFiniteSAdele (x : ProdAdicCompletions R K) :=
  ∀ v, v ∉ S → x v ∈ v.adicCompletionIntegers K

def finiteSAdeleRing : Subring (ProdAdicCompletions R K) where
  carrier := {x | IsFiniteSAdele S x} ...
```

► **Theorem 4.6.** *The finite S -adele ring $\mathbb{A}_{K,S,f}$ of a number field K is locally compact.*

Proof. The finite S adele ring is the product

$$\prod_{v \in S} K_v \times \prod_{v \notin S} \mathcal{O}_v,$$

whose first factor is a finite product of locally compact spaces, hence locally compact, and whose second factor is an infinite product of compact spaces, hence locally compact. ◀

The above proof is typical of the literature, but it hides a some details that are trivial mathematically, but not formally. This comes down to two concerns: (1) how we formalise the space $\prod_{v \in S} K_v \times \prod_{v \notin S} \mathcal{O}_v$; (2) the sense in which we identify $\mathbb{A}_{K,S,f}$ and $\prod_{v \in S} K_v \times \prod_{v \notin S} \mathcal{O}_v$.

For (1), $\prod_{v \in S} K_v \times \prod_{v \notin S} \mathcal{O}_v$ can be formalised as a `Pi` type, given a `Pi` topological space instance, which we denote mathematically by $\prod_{v \in S} K_v \times \prod_{v \notin S} \mathcal{O}_v$. Or, it can be a subtype of $\prod_{v \in S} K_v \times \prod_{v \notin S} K_v$, given a `Subtype` topological space instance, which we denote mathematically by $\mathbb{A}'_{K,S,f}$. We shall use both, because the former is really the space whose local compactness is deduced in the proof of Theorem 4.6 above, but the latter simplifies the transfer of this local compactness to $\mathbb{A}_{K,S,f}$ considerably. The spaces $\prod_{v \in S} K_v \times \prod_{v \notin S} K_v$, $\prod_{v \in S} K_v \times \prod_{v \notin S} \mathcal{O}_v$ and $\mathbb{A}'_{K,S,f}$ are formalised respectively as follows.

```
def SProdAdicCompletions :=
  ((v : S) → v.val.adicCompletion K) × ((v : {v // v ∉ S}) →
  v.val.adicCompletion K)

def SProdAdicCompletionIntegers :=
  ((v : S) → v.val.adicCompletionIntegers K) × ((v : {v // v ∉ S}) →
  v.val.adicCompletionIntegers K)

def SProdAdicCompletionIntegers_subtype := {x : SProdAdicCompletions R K S //
  ∀ v : {v // v ∉ S}, x.2 v ∈ v.val.adicCompletionIntegers K}
```

The identification concern of (2) can then be achieved through composing two homeomorphisms t_1 and t_2 , as illustrated in Figure 1. Here, $p : \prod_v K_v \rightarrow \mathbf{Prop}$; $x \mapsto \forall v \notin S, x_v \in \mathcal{O}_v$ is the subtype condition defining $\mathbb{A}_{K,S,f}$ and $q : (x_1, x_2) \mapsto \forall v \notin S, (x_2)_v \in \mathcal{O}_v$ is the subtype condition defining $\mathbb{A}'_{K,S,f}$. The *partial function* $\langle \cdot, p \rangle : \prod_v K_v \rightarrow \mathbb{A}_{K,S,f}$ is the function defined on each $x \in \prod_v K_v$ satisfying $p(x)$, sending it to the corresponding element in the subtype $\mathbb{A}_{K,S,f}$, and $\langle \cdot, q \rangle$ is defined analogously. The homeomorphism t_π is given in `mathlib` as `Homeomorph.piEquivPiSubtypeProd`.

$$\begin{array}{ccc}
\prod_v K_v & \xleftarrow[t_\pi]{\cong} & \prod_{v \in S} K_v \times \prod_{v \notin S} K_v & & \prod_{v \in S} K_v \times \prod_{v \notin S} \mathcal{O}_v \\
\downarrow \langle \cdot, p \rangle & & \downarrow \langle \cdot, q \rangle & \nearrow t_2 & \\
\mathbb{A}_{K,S,f} & \xleftarrow[t_1]{\dots\dots\dots} & \mathbb{A}'_{K,S,f} & &
\end{array}$$

■ **Figure 1** Type relationships in the formal proof for the local compactness of $\mathbb{A}_{K,S,f}$.

Given the formalisations of the local compactness of K_v and the compactness of \mathcal{O}_v in Section 4.2.1, the local compactness of $\prod_v \in K_v \times \prod_{v \notin S} \mathcal{O}_v$ can be easily inferred.

```

namespace SProdAdicCompletionIntegers

instance : LocallyCompactSpace (SProdAdicCompletionIntegers R K S) :=
  @Prod.locallyCompactSpace _ _ _ Pi.locallyCompactSpace_of_finite
  Pi.locallyCompactSpace

```

The homeomorphism t_2 is defined by specific `Subtype.mk` maps, whose continuity is relatively straightforward to show. This gives the local compactness of $\mathbb{A}'_{K,S,f}$.

```

theorem homeomorph : SProdAdicCompletionIntegers_subtype R K S  $\simeq_t$ 
  SProdAdicCompletionIntegers R K S where ...

instance : LocallyCompactSpace (SProdAdicCompletionIntegers_subtype R K S) :=
  (Homeomorph.locallyCompactSpace_iff (SProdAdicCompletionIntegers.homeomorph R K
  S)).2 inferInstance

end SProdAdicCompletionIntegers

```

It is easy to see that $p = q \circ t_\pi$ on defined domains, so the homeomorphism t_1 can be lifted from t_π using `Homeomorph.subtype` and used to infer that $\mathbb{A}_{K,S,f}$ is locally compact.

```

namespace FiniteSAdeleRing

theorem homeomorph_piSubtypeProd :
  finiteSAdeleRing R K S  $\simeq_t$  SProdAdicCompletionIntegers_subtype R K S :=
  Homeomorph.subtype (Homeomorph.piEquivPiSubtypeProd _ _) <| fun _ =>
  ⟨fun hx v => hx v.1 v.2, fun hx v hv => hx ⟨v, hv⟩⟩

theorem locallyCompactSpace : LocallyCompactSpace (finiteSAdeleRing R K S) :=
  (Homeomorph.locallyCompactSpace_iff (homeomorph_piSubtypeProd R K S)).2
  inferInstance

```

4.2.3 Using the finite S -adele rings to cover the finite adèle ring

► **Theorem 4.7.** *The finite adèle ring $\mathbb{A}_{K,f}$ of a number field K is locally compact.*

Proof. Let $x \in \mathbb{A}_{K,f}$ and N be a neighbourhood of x . We have $x \in \mathbb{A}_{K,S_x,f}$, where S_x is the set of finite places for which $x \notin \mathcal{O}_v$. Since $\mathbb{A}_{K,S_x,f}$ is a member of the generating set

of the restricted product topology, it is open in $\mathbb{A}_{K,f}$. Moreover, it is locally compact by Theorem 4.6. The neighbourhood $N \cap \mathbb{A}_{K,S_x,f}$ of x inside the finite S -adele ring therefore contains a compact neighbourhood M of x . The inclusion map $\mathbb{A}_{K,S_x,f} \hookrightarrow \mathbb{A}_{K,f}$ is continuous open, so the image of M into $\mathbb{A}_{K,f}$ is a compact neighbourhood of x in $\mathbb{A}_{K,f}$. ◀

The subtlety here is that, in Theorem 4.6, we showed that the finite S -adele ring with the subspace topology of $\prod_v K_v$ is locally compact, whereas in the above proof, we are viewing it as a subspace of $\mathbb{A}_{K,f}$. The key step to formalise here is that these topologies coincide. The $\mathbb{A}_{K,f}$ -subspace topology is defined by inducing through the embedding $\mathbb{A}_{K,S,f} \hookrightarrow \mathbb{A}_{K,f}$.

```
def toFiniteAdeleRing : finiteSAdeleRing R K S →+* finiteAdeleRing R K where ...

local notation "e" => toFiniteAdeleRing R K

def adelicTopologicalSpace : TopologicalSpace (finiteSAdeleRing R K S) :=
  TopologicalSpace.induced (e S) (TopologicalSpace.generateFrom
    (FiniteAdeleRing.generatingSet R K))
```

We show equality of topologies via equality of their generating sets.

```
def adelicGeneratingSet : Set (Set (finiteSAdeleRing R K S)) :=
  Set.preimage (e S) '' (FiniteAdeleRing.generatingSet R K)

theorem adelicGenerateFrom : adelicTopologicalSpace R K S =
  TopologicalSpace.generateFrom (FiniteSAdeleRing.adelicGeneratingSet R K S) := by
  rw [adelicGeneratingSet, ← induced_generateFrom_eq]; rfl

theorem adelicGeneratingSet_eq : adelicGeneratingSet R K S =
  Set.preimage Subtype.val '' (
    {U | ∃ (V : (v: HeightOneSpectrum R) → Set (v.adicCompletion K))
      (I : Finset (HeightOneSpectrum R)),
      (∀ v ∈ I, IsOpen (V v)) ∧ U = Set.pi (I.toSet) V}) := ...

theorem topologicalSpace_eq_adelicTopologicalSpace :
  adelicTopologicalSpace R K S = instTopologicalSpaceSubtype := ...

end FiniteSAdeleRing
```

We then show that the embedding $\mathbb{A}_{K,S,f} \hookrightarrow \mathbb{A}_{K,f}$ is an open topological embedding, which we can use to pull back and push forwards open and compact sets as in the proof of Theorem 4.7.

```
theorem toFiniteAdeleRing_openEmbedding : OpenEmbedding (e S) := ...

namespace FiniteAdeleRing

theorem locallyCompactSpace : LocallyCompactSpace (finiteAdeleRing R K) :=
  refine LocallyCompactSpace.mk (fun x N hN => ?_)
  set S := (Filter.eventually_cofinite.1
    ((mem_finiteAdeleRing_iff x.val).1 x.property)).toFinset
  have hx : x.val ∈ finiteSAdeleRing R K S :=
    fun v hv => by rwa [Set.Finite.mem_toFinset, Set.nmem_setOf_iff, not_not] at hv
  obtain ⟨U, hU0, hU1, hU2⟩ := mem_nhds_iff.1 hN
  have hU_S : (e S)-1 U ∈ nhds ⟨x, hx⟩ :=
    mem_nhds_iff.2 ⟨_, subset_rfl,
      (toFiniteAdeleRing_openEmbedding R K S).continuous.isOpen_preimage _ hU1, hU2⟩
  obtain ⟨N_S, hN_S0, hN_S1, hN_S2⟩ :=
```

```

(FiniteSAdeleRing.locallyCompactSpace R K S).local_compact_nhds ⟨x, hx⟩ _ hU_S
refine ⟨(e S) ’’ N_S, ?_, subset_trans (Set.image_subset_iff.2 hN_S1) hU0,
  ((openEmbedding_iff _) .1 (toFiniteAdeleRing_openEmbedding R K
    S)).1.isCompact_iff.1 hN_S2)
obtain ⟨V, hV, hVOpen, hxV⟩ := mem_nhds_iff.1 hN_S0
exact mem_nhds_iff.2 <| ⟨(e S) ’’ V,
  (Set.image_subset_image_iff (toFiniteAdeleRing_openEmbedding R K S).inj).2 hV,
  (toFiniteAdeleRing_openEmbedding R K S).isOpenMap _ hVOpen, ⟨_, hxV, rfl⟩)
end DedekindDomain.FiniteAdeleRing

```

4.3 Local compactness of the adèle ring

```
namespace NumberField.AdeleRing
```

► **Theorem 4.8.** *The adèle ring \mathbb{A}_K of a number field K is locally compact.*

Proof. As the product of the infinite adèle ring and the finite adèle ring, which are each locally compact (Theorems 4.2 and 4.7, respectively), the adèle ring is also locally compact. ◀

```

theorem locallyCompactSpace (K : Type*) [Field K] [NumberField K] :
  LocallyCompactSpace (adeleRing K) := by
  haveI := InfiniteAdeleRing.locallyCompactSpace K
  haveI := FiniteAdeleRing.locallyCompactSpace (ringOfIntegers K) K
  exact Prod.locallyCompactSpace _ _
end NumberField.AdeleRing

```

5 Discussion

5.1 Infinite completions of number fields and software design

In Section 3.1.2, we described the formalisation of the completion of a number field at an infinite place. The approach of that section is a correction to a prior approach that we give here in order to highlight how deficiencies can arise in ways that are not obvious from a mathematical perspective.

Recall that the uniform completion K_v is a field if (K, \mathcal{U}_v) is a completable topological field. The approach we now describe was inspired by the fact that, through the `mathlib` instance `Subfield.completableTopologicalField`, a `Subfield \mathbb{C}` term is always a completable topological field. Therefore, we can use the embedding $\sigma : K \hookrightarrow \mathbb{C}$ to interpret K as a `Subfield \mathbb{C}` term, and apply the uniform completion functor to this term to obtain K'_v .

```

def NumberField.InfinitePlace.subfield
  {K : Type*} [Field K] [NumberField K] (v : InfinitePlace K) : Subfield  $\mathbb{C}$  where
  toSubring := v.embedding.range ...

def NumberField.InfinitePlace.toSubfield : K →+* v.subfield K where ...

instance NumberField.InfinitePlace.subfield_uniformSpace : UniformSpace
  (v.subfield K) := instUniformSpaceSubtype

```

```
def NumberField.InfinitePlace.completion' := (subfield_uniformSpace K v).Completion
instance : Field (v.completion' K) := UniformSpace.Completion.instField
```

The completion K'_v of this section and K_v of Section 3.1.2 are isomorphic as uniform spaces by Theorem 2.10, because their constructions both define abstract completions of K . The canonical norm on K'_v is the *complex* absolute value, but recall that elements of K are first embedded into \mathbb{C} using σ before completion, so that the value of some $x \in K$ in K'_v is still $|\sigma(x)|_{\mathbb{C}}$. Formally, this is guaranteed by defining the coercion appropriately.

```
instance : Coe K (v.completion' K) where
  coe := (UniformSpace.Completion.coe' _) o v.toSubfield K
```

We can show that K'_v is a locally compact space in the same way we did in Section 3.1.2. Since this approach means there is less repetition of code for establishing instances, it initially appears to be a good candidate for formalisation both from a mathematical and software design perspective. However, deficiencies appear when considering the impact of this approach on further usage and development, particularly as a part of `mathlib`. One example of this is that any abstract completion $(Y, \iota : K \rightarrow Y)$ of K comes with the property that, for any uniform continuous homomorphism $f : K \rightarrow Z$ from K to a complete separated uniform field Z , there exists an extension homomorphism $\hat{f} : Y \rightarrow Z$ such that $\hat{f}(\iota(x)) = f(x)$ for all $x \in K$. In `mathlib` this extension can be obtained from the `AbstractCompletion` class, and for uniform completions specifically this is implemented in `UniformSpace.Completion.extensionHom`. For K'_v , uniform completion is only applied to the `Subfield` \mathbb{C} term, so we can readily extend functions from `v.subfield K` but not from K without further work.

Generally, in bypassing the use of `UniformSpace.Completion` directly on K , we sever access to any current or future elements of the uniform completion API that would transfer properties and objects defined on K to the completion K'_v . These will still hold mathematically since the construction defines an abstract completion, but their formalisations will need to be re-implemented, either through the use of the `AbstractCompletion` class or directly. This represents a design deficiency for future development and usage, which would engender further work when maintaining this work as a part of `mathlib`. It is important then to design formalisation research with an eye on both mathematical validity and software design.

5.2 Future work

The local compactness of the adèle ring was used in Tate's thesis [10] to define adelic Dirichlet L -functions and prove their functional equations, thereby establishing the foundations of the Langlands program for $GL(1)$. Some of the theoretical ingredients for Tate's thesis, such as Haar measures [11], are now available in `mathlib`. This work represents a step towards the formalisation of Tate's thesis in `mathlib`. We list some future work to follow on from this, in order of immediacy.

- Show that the adèle ring of a function field is locally compact.
- Show that K is a discrete cocompact subgroup of \mathbb{A}_K .
- Prove that the idele group \mathbb{I}_K is locally compact and K^\times is a discrete cocompact subgroup of \mathbb{I}_K .
- Formalise adelic Hecke characters.
- Formalise Tate's thesis.

References

- 1 Nicolas Bourbaki. *Elements of mathematics. General topology. Part 1*. Hermann, Paris; Addison-Wesley Publishing Co., Reading, Mass.-London-Don Mills, Ont., 1966.
- 2 Kevin Buzzard, Johan Commelin, and Patrick Massot. Formalising perfectoid spaces. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs, CPP 2020*, pages 299–312, New York, NY, USA, 2020. Association for Computing Machinery. doi:10.1145/3372885.3373830.
- 3 John William Scott Cassels and Albrecht Fröhlich. *Algebraic number theory*. Academic Press, London; Thompson Book Co., Inc., Washington, D.C., 1967.
- 4 María Inés de Frutos-Fernández. Formalizing the Ring of Adèles of a Global Field. In June Andronick and Leonardo de Moura, editors, *13th International Conference on Interactive Theorem Proving (ITP 2022)*, volume 237 of *Leibniz International Proceedings in Informatics (LIPIcs)*, pages 14:1–14:18, Dagstuhl, Germany, 2022. Schloss Dagstuhl – Leibniz-Zentrum für Informatik. URL: <https://drops.dagstuhl.de/opus/volltexte/2022/16723>, doi:10.4230/LIPIcs.ITP.2022.14.
- 5 María Inés de Frutos-Fernández and Filippo A. E. Nuccio Mortarino Majno di Capriglio. A formalization of complete discrete valuation rings and local fields. In *Proceedings of the 13th ACM SIGPLAN International Conference on Certified Programs and Proofs, CPP 2024*, pages 190–204, New York, NY, USA, 2024. Association for Computing Machinery. doi:10.1145/3636501.3636942.
- 6 The mathlib Community. The lean mathematical library. In *Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs, POPL '20*. ACM, January 2020. URL: <http://dx.doi.org/10.1145/3372885.3373824>, doi:10.1145/3372885.3373824.
- 7 Alexander Ostrowski. Über einige Lösungen der Funktionalgleichung $\psi(x) \cdot \psi(x) = \psi(xy)$. *Acta Mathematica*, 41(none):271 – 284, 1916. doi:10.1007/BF02422947.
- 8 Xavier Roblot. leanprover-community/mathlib#17844: add infinite_places. GitHub, 2022. URL: <https://github.com/leanprover-community/mathlib/pull/17844>.
- 9 Jean-Pierre Serre. *Local Fields*. Graduate Texts in Mathematics. Springer New York, 1979. doi:<https://doi.org/10.1007/978-1-4757-5673-9>.
- 10 John T. Tate. Fourier analysis in number fields, and hecke’s zeta-functions. In *Algebraic Number Theory*, Proc. Instructional Conf., Brighton, 1965, 1950. URL: <https://api.semanticscholar.org/CorpusID:117898073>.
- 11 Floris van Doorn. Formalized Haar Measure. In Liron Cohen and Cezary Kaliszyk, editors, *12th International Conference on Interactive Theorem Proving (ITP 2021)*, volume 193 of *Leibniz International Proceedings in Informatics (LIPIcs)*, pages 18:1–18:17, Dagstuhl, Germany, 2021. Schloss Dagstuhl – Leibniz-Zentrum für Informatik. URL: <https://drops.dagstuhl.de/entities/document/10.4230/LIPIcs.ITP.2021.18>, doi:10.4230/LIPIcs.ITP.2021.18.
- 12 A. Weil. *Sur les espaces à structure uniforme et sur la topologie générale*. Actualités scientifiques et industrielles. Hermann & cie, 1938. URL: <https://books.google.co.uk/books?id=qVfvAAAAAAAJ>.